

Program Title CURVE FITTING

Contributor's Name KEITH JARETT

Address

City

State/Country

Zip Code

Program Description, Equations, Variables

This program fits data to the equation $Y=AX+B$, where X and Y are functions of x and y respectively. The least squares fit is:

$$A = (\sum x_i y_i - \sum x_i \sum y_i / n) / (\sum x_i^2 - (\sum x_i)^2 / n) ; B = (\sum y_i - A \sum x_i) / n ;$$

$$r^2 = (\sum x_i y_i - \sum x_i \sum y_i / n)^2 / (\sum x_i^2 - (\sum x_i)^2 / n) (\sum y_i^2 - (\sum y_i)^2 / n) , \text{ coefficient of determination ;}$$

$$r = (\text{sign } A) \sqrt{r^2} , \text{ correlation coefficient.}$$

Four types of curves can be fit:

$j=1$, linear: $y=ax+b$

$j=2$, exponential: $y=b e^{ax}$; $Y=\ln y$, $A=a$, $X=x$, $B=\ln b$.

$j=3$, logarithmic: $y=a \ln x + b$; $Y=y$, $A=a$, $X=\ln x$, $B=b$.

$j=4$, power curve: $y=b x^a$; $Y=\ln y$, $A=a$, $X=\ln x$, $B=\ln b$.

For the user's convenience, the program provides progressive integer values which can be used for x coordinates.

Necessary Accessories none

Operating Limits and Warnings x or y values $\leq$ zero will not work, since the calculator needs to take logarithms. Use 10^{-9} instead of zero.

If A is exactly zero, a DATA ERROR will occur. Thus all y_i 's cannot be the same. Similarly all x_i 's cannot be the same ($A \neq 00$).

Reference(s) HP-35 Math Pac, page 144.

Curve Fitting-- Made Easy, by Gary M. Tenzer, PPC Journal Jan 78. Gary's program for the HP67/97 provide the framework and user-oriented features for this program.

The program flow has been improved, the code compacted, and the data storage requirement decreased.

This program has been verified only with respect to the numerical example given in Program Description II. User accepts and uses this program material AT HIS OWN RISK, in reliance solely upon his own inspection of the program material and without reliance upon any representation or description concerning the program material.

NEITHER HP NOR THE CONTRIBUTOR MAKES ANY EXPRESS OR IMPLIED WARRANTY OF ANY KIND WITH REGARD TO THIS PROGRAM MATERIAL, INCLUDING, BUT NOT LIMITED TO, THE IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE. NEITHER HP NOR THE CONTRIBUTOR SHALL BE LIABLE FOR INCIDENTAL OR CONSEQUENTIAL DAMAGES IN CONNECTION WITH OR ARISING OUT OF THE FURNISHING, USE OR PERFORMANCE OF THIS PROGRAM MATERIAL.

Sample Problem (Sketch if Desired)

A. Fit a curve to the following data:

X	1	2	3	3.5	4
Y	.45	4.2	14.2	22.7	43.1

B. Repeat after substituting (2,3.2) for (2,4.2).

C. Find the best exponential curve fit to the data in (B).

D. Calculate the Y estimate for $x=5$.

SOLUTION:

Input	Function	Display	Comments
	[USER]		
	[XEQ] SIZE 018		
	■ [A]	1.00	Initialize
A. .45	R/S	2.00	Previous output used for x.
4.5	R/S	3.00	Whoops! Wrong entry.
2	ENTER ↑		
4.5	■ [C]	2.00	Cancel incorrect entry.
4.2	R/S	3.00	Enter correct value.
14.2	R/S	4.00	
43.1	R/S	5.00	
3.5	ENTER ↑		
22.7	R/S	6.00	
	[A]	1.00	Flag 4 set (power curve)
	FIX 4	0.9979	r^2
	R↓	3.2113	a
	R↓	0.4446	b (y=bx ^a)
B. 2	ENTER ↑		Cancel incorrect value.
4.2	■ [E]	5.0000	
2	ENTER ↑		Enter correct value.
3.2	[E]	6.0000	
	[A]	0.9945	r^2 Flag 4 set again.
	R↓	3.2532	a
	R↓	0.4057	b (y=bx ^a)
C. 2	[B]	0.9859	r^2 Flag 2 set (exponential)
	R↓	1.5051	a
	R↓	0.1245	b (y=be ^{ax})
D. 5	[D]	231.0266	$\hat{y}$ for x=5.

				SIZE: (HP-41C) 018
STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Set USER mode and sufficient SIZE			
2.	Initialize		 [A]	1.00
	1.00 can be used for the first x value.			
3.	Enter data for $i=1,2,\dots,n$.	x_i	ENTER $\uparrow$	
	$i+1$ can be used for the next x value.	y_i	[E] or R/S	$i+1$
4.	Erroneous input correction:			
	Re-enter incorrect values	x_i	ENTER $\uparrow$	
		y_i	 [E]	i
	Enter correct values	x_i	ENTER $\uparrow$	
		y_i	[E] or R/S	$i+1$
	Skip to step 6 if a specific curve type is desired.			
5.	Find best curve type		[A]	Flag j on
	(j=1 for linear, 2 for exponential,			r^2
	3 for logarithmic, 4 for power curve)		R $\downarrow$	a
			R $\downarrow$	b
6.	Fit specific curve type	j	[B]	Flag j on
	(j=1 for linear, 2 for exponential,			r^2
	3 for logarithmic, 4 for power curve)		R $\downarrow$	a
			R $\downarrow$	b
7.	Estimate Y for a given X	x	[D]	$\hat{y}$
8.	Review curve parameters		[C]	r^2
			R $\downarrow$	a
			R $\downarrow$	b
OPTIONAL FEATURE-- See the continuation page				
9.	Calculate correlation coefficient r, where $r = (\text{sign } a) \sqrt{r^2}$.		 [C]	r

(CONTINUATION PAGE)

OPTIONAL FEATURE-- Correlation Coefficient

The following program segment calculates the correlation coefficient r of a curve fit, using the formula $r = (\text{sign } a)\sqrt{r^2}$.

```
182   LBL c
183   RCL 03
184   SQRT
185   RCL 02
186   SIGN
187   *
188   RTN
```

Simply add the above program segment to the bottom of the program if you desire this feature. The total memory requirement will rise by one register, from 054 to 055.

Pressing $\blacksquare$ [C] after a curve fit has been performed (and r^2 , a , and b are known) will give the correlation coefficient r for that curve.

If you can easily spare another register and you need to get correlation coefficients for your curve fits, it makes sense to add this segment to your program.

PROGRAM LISTING

□ 67 □ 97 ☒ 41C

STEP/ LINE	KEY ENTRY	KEY CODE (67/97 only)	COMMENTS	STEP/ LINE	KEY ENTRY	KEY CODE (67/97 only)	COMMENTS
01	LBL "CRV FIT"			47	LN		
02	LBL a		Initialize	48	ΣREG 12		
03	SF 25			49	RTN		
04	CF IND 0			50	LBL A		Solve best
0				51	SF 25		
05	CLRG			52	CF IND 0		Clear old type select flag
06	LBL 09			53	0		
07	RCL 11			54	STO 03		
08	1			55	XEQ 01		linear
09	+			56	XEQ 02		exponential
10	RTN			57	XEQ 03		logarithmic
11	LBL E		Data entry	58	XEQ 04		power curve
12	ΣREG 06			59	GTO 12		
13	STO L			60	LBL B		
14	Σ+		Sum data into R ₀₄	61	CF IND 0		Clear old type select flag
15	XEQ 00			62	STO 00		
16	Σ+		to R ₁₀	63	LBL 12		
17	R↑			64	SF IND 0		Set selected type flag
18	ST+ 05			65	SF 07		
19	R↑			66	XEQ IND		Solve selected type
20	ST+ 04			67	LBL C		Place output data in stack
21	GTO 09			68	RCL 01		
22	LBL e		Data cancel	69	RCL 02		
23	ΣREG 06			70	RCL 03		
24	STO L			71	RTN		
25	Σ-			72	LBL 01		Load stack and R ₀₂ with correct
26	XEQ 00			73	RCL 07		
27	Σ-			74	STO 02		
28	R↑			75	RCL 09		data for linear curve fit.
29	ST- 05			76	RCL 10		
30	R↑			77	RCL 06		
31	ST- 04			78	RCL 08		
32	GTO 09			79	XEQ 05		
33	LBL 00			80	1		
34	RDN			81	GTO 14		
35	LASTX			82	LBL 02		Load data for exponential
36	LN			83	RCL 13		
37	X<>Y			84	STO 02		
38	LASTX			85	RCL 09		
39	X<>Y			86	RCL 05		
40	LN			87	RCL 12		
41	X<>Y			88	RCL 08		
42	LASTX			89	XEQ 05		
43	ST* T			90	2		
44	LN						
45	X<>Y						
46	ST* Z						



□ 67 □ 97 ☒ 41C

STEP/ LINE	KEY ENTRY	KEY CODE (67/97 only)	COMMENTS	STEP/ LINE	KEY ENTRY	KEY CODE (67/97 only)	COMMENTS
91	GTO	14		139	FS?C	07	
92	♦LBL	03	Load data for	140	GTO	06	
93	RCL	07	logarithmic fit	141	RCL	03	
94	STO	02		142	X<>Y		
95	RCL	15		143	X<=Y?		
96	RCL	04		144	RTN		
97	RCL	06		145	SF	06	
98	RCL	14		146	♦LBL	06	
99	XEQ	05		147	STO	03	
100	3			148	RCL	17	
101	GTO	14		149	RCL	01	
102	♦LBL	04	Load data for	150	RCL	02	
103	RCL	13	power curve fit	151	*		
104	STO	02		152	-		
105	RCL	15		153	RCL	11	
106	RCL	16		154	/		
107	RCL	12		155	FS?	02	
108	RCL	14		156	E↑X		
109	XEQ	05		157	FS?	04	
110	4			158	E↑X		
111	♦LBL	14		159	STO	01	
112	FS?C	06		160	RTN		
113	STO	00	Store r^2 if it	161	♦LBL	D	Use a and b values
114	RTN		beats the previous	162	RCL	01	to calculate
115	♦LBL	05	best	163	RCL	02	y estimate for a
116	STO	01	Find and store	164	RCL	Z	given x
117	X<>Y		a and b, then	165	GTO	IND	
118	STO	17	see whether r^2	00			
119	*		is better than	166	♦LBL	03	
120	RCL	11	the previous best	167	LN		
121	/			168	♦LBL	01	
122	-			169	*		
123	X<>Y			170	+		
124	RCL	01		171	RTN		
125	X↑2			172	♦LBL	02	
126	RCL	11		173	*		
127	/			174	E↑X		
128	-			175	*		
129	/			176	RTN		
130	*			177	♦LBL	04	
131	LASTX			178	X<>Y		
132	X<>	02		179	Y↑X		
133	RCL	17		180	*		
134	X↑2			181	RTN		
135	RCL	11		182	.END.		
136	/						
137	-						
138	/						


DATA REGISTERS				STATUS			
00	j (curve type)			SIZE 018	TOT. REG. 54	USER MODE	
01	b ; temporary Σx			ENG	FIX SCI	ON X OFF	
02	a ; temp. Σy^2			DEG	RAD GRAD		
03	r^2			diplay and trig modes as desired			
04	$\Sigma y \ln x$			FLAGS			
05	$\Sigma x \ln y$			#	INIT S/C	SET INDICATES	CLEAR INDICATES
06	Σy						
07	Σy^2			1	C	linear fit	
08	Σx			2	C	exponential fit	
09	Σx^2						
10	Σxy			3	C	logarithmic fit	
11	n			4	C	power curve fit	
12	$\Sigma \ln y$			6	C	update r^2	do not update r^2
13	$\Sigma (\ln y)^2$			7	C	specific type set	find best r^2
14	$\Sigma \ln x$						
15	$\Sigma (\ln x)^2$						
16	$\Sigma (\ln x)(\ln y)$						
17	n ; temp. Σy						
				ASSIGNMENTS			
				FUNCTION	KEY	FUNCTION	KEY
				Optional-- Assign Roll Down (R↓) to its own key (22) to speed its execution in USER mode.			

KEYBOARD CARD LABELING

KEYBOARD

SYSTEM
CONFIGURATION

CARD



PROGRAM REGISTERS NEEDED: 37

ROW 1 (1 : 3)



ROW 2 (3 : 12)



ROW 3 (13 : 20)



ROW 4 (21 : 27)



ROW 5 (28 : 37)



ROW 6 (38 : 48)



ROW 7 (48 : 55)



ROW 8 (56 : 60)



ROW 9 (61 : 68)



ROW 10 (69 : 79)



ROW 11 (80 : 89)



ROW 12 (90 : 99)



ROW 13 (100 : 109)



ROW 14 (109 : 119)



ROW 15 (120 : 132)



ROW 16 (132 : 141)



ROW 17 (142 : 152)



ROW 18 (153 : 162)



ROW 19 (163 : 173)



ROW 20 (174 : 182)

